

# INSTRUCTION MANUAL



## Model HFP01SC Self-Calibrating Soil Heat Flux Plate

Revision: 8/07



Copyright © 2002-2007  
Campbell Scientific, Inc.

# **Warranty and Assistance**

---

The **MODEL HFP01SC SELF-CALIBRATING SOIL HEAT FLUX PLATE** is warranted by CAMPBELL SCIENTIFIC, INC. to be free from defects in materials and workmanship under normal use and service for twelve (12) months from date of shipment unless specified otherwise. Batteries have no warranty. CAMPBELL SCIENTIFIC, INC.'s obligation under this warranty is limited to repairing or replacing (at CAMPBELL SCIENTIFIC, INC.'s option) defective products. The customer shall assume all costs of removing, reinstalling, and shipping defective products to CAMPBELL SCIENTIFIC, INC. CAMPBELL SCIENTIFIC, INC. will return such products by surface carrier prepaid. This warranty shall not apply to any CAMPBELL SCIENTIFIC, INC. products which have been subjected to modification, misuse, neglect, accidents of nature, or shipping damage. This warranty is in lieu of all other warranties, expressed or implied, including warranties of merchantability or fitness for a particular purpose. CAMPBELL SCIENTIFIC, INC. is not liable for special, indirect, incidental, or consequential damages.

Products may not be returned without prior authorization. The following contact information is for US and International customers residing in countries served by Campbell Scientific, Inc. directly. Affiliate companies handle repairs for customers within their territories. Please visit [www.campbellsci.com](http://www.campbellsci.com) to determine which Campbell Scientific company serves your country. To obtain a Returned Materials Authorization (RMA), contact CAMPBELL SCIENTIFIC, INC., phone (435) 753-2342. After an applications engineer determines the nature of the problem, an RMA number will be issued. Please write this number clearly on the outside of the shipping container. CAMPBELL SCIENTIFIC's shipping address is:

**CAMPBELL SCIENTIFIC, INC.**

RMA#\_\_\_\_\_

815 West 1800 North

Logan, Utah 84321-1784

CAMPBELL SCIENTIFIC, INC. does not accept collect calls.

# HFP01SC Table of Contents

---

*PDF viewers note: These page numbers refer to the printed version of this document. Use the Adobe Acrobat® bookmarks tab for links to specific sections.*

<b>1. General Description</b> .....	1
<b>2. Sensor Specifications</b> .....	1
<b>3. Installation</b> .....	2
<b>4. Wiring</b> .....	4
<b>5. Example Programs</b> .....	4
<b>6. Soil Heat Flux and Storage</b> .....	24
<b>7. Maintenance</b> .....	24
<b>8. In-Situ Calibration Theory</b> .....	24
<b>9. References</b> .....	25

## **Figures**

1. Placement of Heat Flux Plates .....	2
2. HFP01SC Plate .....	3

## **Tables**

1. Datalogger Connections for a Single-Ended Measurement.....	3
2. Datalogger Connections for a Differential Measurement .....	4
3. Wiring for Example 1 .....	5
4. Wiring for Example 2 .....	9
5. Wiring for Example 3 .....	14
6A. Wiring for Example 4 .....	17
6B. Wiring for Example 4 .....	17
7. Hukseflux and Campbell Scientific Variable Names .....	25



# **Model HFP01SC**

## **Self-Calibrating Soil Heat Flux Plate**

---

### **1. General Description**

The HFP01SC Soil Heat Flux plate consists of a thermopile and a film heater. The thermopile measures temperature gradients across the plate. During the in-situ field calibration, the film heater is used to generate a heat flux through the plate. The amount of power used to generate the calibration heat flux is measured by the datalogger. Each plate is individually calibrated, at the factory, to output flux.

In order to measure soil heat flux at the surface, several HFP01SCs are used to measure the soil heat flux at a depth of eight centimeters. A TCAV Averaging Soil Thermocouple is used to measure the temporal change in temperature of the soil layer above the HFP01SC. Finally, a CS616 Water Content Reflectometer is used to measure the soil water content. The temporal change in soil temperature and soil water content are used to compute the soil storage term.

The -L option on the model HFP01SC Soil Heat Flux plate (HFP01SC-L) indicates that the cable length is user specified. The HFP01SC-L has two cables; the first cable is the signal output cable and the second is the heater input cable. Two analog inputs are required to measure the HFP01SC-L. This manual refers to the sensor as the HFP01SC.

### **2. Specifications**

Operating Temperature:	-30°C to +70°C
Storage Temperature:	-30°C to +70°C
Plate Thickness:	5 mm (0.2 in.)
Plate Diameter:	80 mm (3.15 in.)
Power Consumption	
During Self Calibration:	1.3 W (108 mA @ 12 vdc)
Sensor:	thermopile and film heater
Measurement Range:	$\pm 100 \text{ W m}^{-2}$
Signal Range (nominal):	$\pm 5 \text{ mV}$ for the above range (sensor), 0 to 2 V (while the film heater is powered)
Accuracy:	$\pm 3\%$ of reading
Sensitivity (nominal):	$67 \mu\text{V W}^{-1} \text{ m}^{-2}$
Sensor Resistance (nominal):	2 $\Omega$
Heater Resistance (nominal):	100 $\Omega$
Current Sensing Resistor:	10 $\Omega$ 1% 0.25 W
Thermal Conductivity:	$0.8 \text{ W m}^{-1} \text{ K}^{-1}$




FIGURE 1. Placement of Heat Flux Plates

### 3. Installation

The HFP01SC Soil Heat Flux plates, the TCAV Averaging Soil Temperature probes, and the CS616 Water Content Reflectometer are installed as shown in Figure 1.

The location of the heat flux plates and thermocouples should be chosen to be representative of the area under study. If the ground cover is extremely varied, it may be necessary to have additional sensors to provide a valid spatial average of soil heat flux.

Use a small shovel to make a vertical slice in the soil. Excavate the soil to one side of the slice. Keep this soil intact so that it can be replaced with minimal disruption.

The sensors are installed in the undisturbed face of the hole. Measure the sensor depths from the top of the hole. With a small knife, make a horizontal cut eight centimeters below the surface into the undisturbed face of the hole. Insert the heat flux plate into the horizontal cut.

---

**NOTE**

Install the HFP01SC in the soil such that the side with the text “this side up” is facing the sky.

---

**CAUTION**

In order for the HFP01SC to make quality soil heat flux measurements, the plate must be in full contact with the soil.

Never run the sensors leads directly to the surface. Rather, bury the sensor leads a short distance back from the hole to minimized thermal conduction on the lead wire. Replace the excavated soil back into its original position after all the sensors are installed.




FIGURE 2. HFP01SC Plate

TABLE 1. Datalogger Connections for a Single-Ended Measurement			
Description	Color	CR10X	CR23X & CR5000
Sensor Signal	White	Single-Ended Input	Single-Ended Input
Sensor Signal Reference	Green	AG	⚡
Shield	Clear	G	⚡
Heater Resistor Signal	Yellow	Single-Ended Input	Single-Ended Input
Heater Resistor Signal Reference	Purple	AG	⚡
Shield	Clear	G	⚡
Power	Red	SW12	SW12
Power Reference	Black	G	G
External Power Control	jumper wire	SW12-CTRL to Control Port	External Power Control Not Needed

Description	Color	CR10(X)	CR23X & CR5000
Sensor Signal	White	Differential Input (H)	Differential Input (H)
Sensor Signal Reference	Green	Differential Input (L)	Differential Input (L)
Shield	Clear	G	⊕
Heater Resistor Signal	Yellow	Differential Input (H)	Differential Input (H)
Heater Resistor Signal Reference	Purple	Differential Input (L)	Differential Input (L)
Shield	Clear	G	⊕
Power	Red	SW12	SW12
Power Reference	Black	G	G
External Power Control	Jumper wire	SW12-CTRL to Control Port	External Power Control Not Needed

## 4. Wiring

Connections to Campbell Scientific dataloggers are given in Tables 1 and 2. The output of the HFP01SC can be measured using a single-ended analog measurement (Instruction 1 or VoltSE()), however, a differential analog measurement (Instruction 2 or VoltDiff()) is recommended.

The wiring convention is that the white wire is positive with respect to the green wire, when energy is flowing through the transducer from the side with the text “this side up” to the other side.

**NOTE**

The switched 12 vdc port on the black CR10X wiring panel can source enough current to calibrate five HFP01SC plates; the CR23X and CR5000 switched 12 vdc port can be used to calibrate four HFP01SC plates. If additional HFP01SC plates are needed an external relay is required to power the additional plates (see example 4).

For dataloggers without a SW12V output (CR7X, 21X and CR10), a relay (A21REL-12) is required for the in-situ calibration (see Example 4).

## 5. Example Programs

The HFP01SC has a nominal calibration of  $15 \text{ W m}^{-2} \text{ mV}^{-1}$ . Each sensor is accompanied by a calibration certificate. Each sensor also has a unique calibration label on it. The label is located on the pigtail end of the sensor leads.

**TABLE 3. Wiring for Example 1**

Description	Color	CR10(X)
Sensor Signal	White	IH
Sensor Signal Reference	Green	AG
Shield	Clear	G
Heater Resistor Signal	Yellow	IL
Heater Resistor Signal Reference	Purple	AG
Shield	Clear	G
Power	Red	SW12
Power Reference	Black	G
External Power Control		jumper wire SW12-CTRL to C8

**Example 1. Sample CR10(X) Program Using a Single-Ended Measurement Instruction**

```

;{CR10X}
;
*Table 1 Program
01: 1          Execution Interval (seconds)

;Measure HFP01SC on smaller range.
;
1: Volt (SE) (P1)
1: 1          Reps
2: 22         7.5 mV 60 Hz Rejection Range
3: 1          SE Channel
4: 2          Loc [ shf_mV  ]
5: 1          Mult
6: 0          Offset

;Measure HFP01SC on larger range.
;
2: Volt (SE) (P1)
1: 1          Reps
2: 23         25 mV 60 Hz Rejection Range
3: 1          SE Channel
4: 8          Loc [ shf_mV_a ]
5: 1          Mult
6: 0          Offset

;Load in the factory calibration.
;
3: If (X<=>F) (P89)
1: 3          X Loc [ cal  ]
2: 1          =
3: 0          F
4: 30         Then Do
    
```

```

;Factory calibration in W/(m^2 mV) = 1000/sensitivity.
;
4: Z=F (P30)
  1: 1      F      ; <- Enter the unique calibration here
  2: 0      Exponent of 10
  3: 3      Z Loc [ cal   ]

5: End (P95)

;Use data from the larger measurement range.
;
6: If (X<=>F) (P89)
  1: 2      X Loc [ shf_mV  ]
  2: 4      <
  3: -99990 F
  4: 30     Then Do

7: Z=X (P31)
  1: 8      X Loc [ shf_mV_a ]
  2: 2      Z Loc [ shf_mV  ]

8: End (P95)

;Apply custom calibration to the raw soil heat flux measurement.
;
9: Z=X*Y (P36)
  1: 2      X Loc [ shf_mV  ]
  2: 3      Y Loc [ cal   ]
  3: 1      Z Loc [ shf   ]

;Output data.
;
10: If time is (P92)
  1: 0      Minutes (Seconds --) into a
  2: 20     Interval (same units as above)
  3: 10     Set Output Flag High (Flag 0)

11: Real Time (P77)
  1: 0110   Day,Hour/Minute (midnight = 0000)

12: Resolution (P78)
  1: 1      High Resolution

;Do not include the calibration data in the soil heat flux.
;
13: If Flag/Port (P91)
  1: 18     Do if Flag 8 is High
  2: 19     Set Intermed. Proc. Disable Flag High (Flag 9)

14: Average (P71)
  1: 1      Reps
  2: 1      Loc [ shf   ]

15: Do (P86)
  1: 29     Set Intermed. Proc. Disable Flag Low (Flag 9)

```

```

16: Sample (P70)
   1: 1      Reps
   2: 3      Loc [ cal  ]

;Add other processing here.

;Call calibration routine.
;
17: Do (P86)
   1: 8      Call Subroutine 8

*Table 2 Program
   02: 0      Execution Interval (seconds)

*Table 3 Subroutines

;Calibration routine.
;
1: Beginning of Subroutine (P85)
   1: 8      Subroutine 8

;Perform in-situ calibration.
;
2: If time is (P92)
   1: 1      Minutes (Seconds --) into a
   2: 180    Interval (same units as above)
   3: 30     Then Do

3: Z=X (P31)
   1: 2      X Loc [ shf_mV  ]
   2: 4      Z Loc [ mV_0  ]

;Begin heating for calibration.
;
4: Do (P86)
   1: 48     Set Port 8 High

;Used to filter data during and after calibration.
;
5: Do (P86)
   1: 18     Set Flag 8 High

6: End (P95)

;End site calibration three minutes after calibration started.
;
7: If time is (P92)
   1: 4      Minutes (Seconds --) into a
   2: 180    Interval (same units as above)
   3: 30     Then Do

```

```

;Measure voltage across current shunt resistor (10 ohm 1% 0.25 W 50
;ppm/deg C) during calibration. This measurement is used to
;compute power.
;
8: Volt (SE) (P1)
  1: 1      Repts
  2: 25     2500 mV 60 Hz Rejection Range
  3: 2      SE Channel
  4: 7      Loc [ V_Rf   ]
  5: .001   Mult
  6: 0      Offset

9: Z=X (P31)
  1: 2      X Loc [ shf_mV  ]
  2: 5      Z Loc [ mV_180  ]

;Turn off the soil heat flux plate heater.
;
10: Do (P86)
  1: 58     Set Port 8 Low

11: End (P95)

;Stop filtering data.
;
12: If time is (P92)
  1: 39     Minutes (Seconds --) into a
  2: 180    Interval (same units as above)
  3: 30     Then Do

13: Do (P86)
  1: 28     Set Flag 8 Low

;Compute in-situ calibration.
;
14: Z=X (P31)
  1: 2      X Loc [ shf_mV  ]
  2: 6      Z Loc [ mV_end  ]

15: Z=X*Y (P36)
  1: 7      X Loc [ V_Rf   ]
  2: 7      Y Loc [ V_Rf   ]
  3: 3      Z Loc [ cal    ]

16: Z=X*F (P37)
  1: 3      X Loc [ cal    ]
  2: 128.7  F
  3: 3      Z Loc [ cal    ]

17: Z=X+Y (P33)
  1: 4      X Loc [ mV_0    ]
  2: 6      Y Loc [ mV_end  ]
  3: 9      Z Loc [ work    ]

18: Z=X*F (P37)
  1: 9      X Loc [ work    ]
  2: .5     F
  3: 9      Z Loc [ work    ]

```

```

19: Z=X-Y (P35)
   1: 9      X Loc [ work  ]
   2: 5      Y Loc [ mV_180 ]
   3: 9      Z Loc [ work  ]

20: Z=ABS(X) (P43)
   1: 9      X Loc [ work  ]
   2: 9      Z Loc [ work  ]

21: Z=X/Y (P38)
   1: 3      X Loc [ cal   ]
   2: 9      Y Loc [ work  ]
   3: 3      Z Loc [ cal   ]

22: End (P95)

23: End (P95)

End Program

-Input Locations-
1 shf
2 shf_mV
3 cal
4 mV_0
5 mV_180
6 mV_end
7 V_Rf
8 shf_mV_a
9 work
    
```

**TABLE 4. Wiring for Example 2**

Description	Color	CR23X
Sensor Signal	White	9H
Sensor Signal Reference	Green	9L
Shield	Clear	≠
Heater Resistor Signal	Yellow	10H
Heater Resistor Signal Reference	Purple	10L
Shield	Clear	≠
Power	Red	SW12
Power Reference	Black	G

**Example 2. Sample CR23X Program Using a Differential Measurement Instruction**

```

;{CR23X}
;
*Table 1 Program
  01: 1      Execution Interval (seconds)

;Measure HFP01SC on smaller range.
;
1: Volt (Diff) (P2)
  1: 1      Reps
  2: 21     10 mV, 60 Hz Reject, Slow Range
  3: 9      DIFF Channel
  4: 2      Loc [ shf_mV  ]
  5: 1      Mult
  6: 0      Offset

;Measure HFP01SC on larger range.
;
2: Volt (Diff) (P2)
  1: 1      Reps
  2: 25     5000 mV, 60 Hz Reject, Fast Range
  3: 9      DIFF Channel
  4: 8      Loc [ shf_mV_a ]
  5: 1      Mult
  6: 0      Offset

;Load in the factory calibration.
;
3: If (X<=>F) (P89)
  1: 3      X Loc [ cal  ]
  2: 1      =
  3: 0      F
  4: 30     Then Do

;Factory calibration in W/(m^2 mV) = 1000/sensitivity.
;
4: Z=F (P30)
  1: 1      F ; <- Enter the unique calibration here
  2: 0      Exponent of 10
  3: 3      Z Loc [ cal  ]

5: End (P95)

;Use data from the larger measurement range.
;
6: If (X<=>F) (P89)
  1: 2      X Loc [ shf_mV  ]
  2: 4      <
  3: -99990 F
  4: 30     Then Do

7: Z=X (P31)
  1: 8      X Loc [ shf_mV_a ]
  2: 2      Z Loc [ shf_mV  ]

8: End (P95)

```

```

;Apply custom calibration to the raw soil heat flux measurement.
;
9: Z=X*Y (P36)
  1: 2      X Loc [ shf_mV  ]
  2: 3      Y Loc [ cal    ]
  3: 1      Z Loc [ shf    ]

;Output data.
;
10: If time is (P92)
  1: 0      Minutes (Seconds --) into a
  2: 20     Interval (same units as above)
  3: 10     Set Output Flag High (Flag 0)

11: Real Time (P77)
  1: 0110   Day,Hour/Minute (midnight = 0000)

12: Resolution (P78)
  1: 1      High Resolution

;Do not include that calibration data in the soil heat flux.
;
13: If Flag/Port (P91)
  1: 118    Do if Flag 18 is High
  2: 19     Set Intermed. Proc. Disable Flag High (Flag 9)

14: Average (P71)
  1: 1      Reps
  2: 1      Loc [ shf    ]

15: Do (P86)
  1: 29     Set Intermed. Proc. Disable Flag Low (Flag 9)

16: Sample (P70)
  1: 1      Reps
  2: 3      Loc [ cal    ]

;Add other processing here.

;Call calibration routine.
;
17: Do (P86)
  1: 8      Call Subroutine 8

*Table 2 Program
  02: 0     Execution Interval (seconds)

*Table 3 Subroutines

;Calibration routine.
;
1: Beginning of Subroutine (P85)
  1: 8      Subroutine 8

;Perform in-situ calibration.
;

```

```

2: If time is (P92)
  1: 1      Minutes (Seconds --) into a
  2: 180    Interval (same units as above)
  3: 30     Then Do

3: Z=X (P31)
  1: 2      X Loc [ shf_mV  ]
  2: 4      Z Loc [ mV_0   ]

;Begin heating for calibration.
;
4: Do (P86)
  1: 49     Turn On Switched 12V

;Used to filter data during and after calibration.
;
5: Do (P86)
  1: 118    Set Flag 18 High

6: End (P95)

;End site calibration three minutes after calibration started.
;
7: If time is (P92)
  1: 4      Minutes (Seconds --) into a
  2: 180    Interval (same units as above)
  3: 30     Then Do

;Measure voltage across current shunt resistor during calibration.
;This measurement is used to compute power.
;
8: Volt (Diff) (P2)
  1: 1      Repts
  2: 25     5000 mV, 60 Hz Reject, Fast Range
  3: 10     DIFF Channel
  4: 7      Loc [ V_Rf    ]
  5: .001   Mult
  6: 0      Offset

9: Z=X (P31)
  1: 2      X Loc [ shf_mV  ]
  2: 5      Z Loc [ mV_180  ]

;Turn off the soil heat flux plate heater.
;
10: Do (P86)
  1: 59     Turn Off Switched 12V

11: End (P95)

;Stop filtering data.
;
12: If time is (P92)
  1: 39     Minutes (Seconds --) into a
  2: 180    Interval (same units as above)
  3: 30     Then Do

```

```

13: Do (P86)
  1: 218          Set Flag 18 Low

;Compute in-situ clibration.
;
14: Z=X (P31)
  1: 2           X Loc [ shf_mV  ]
  2: 6           Z Loc [ mV_end  ]

15: Z=X*Y (P36)
  1: 7           X Loc [ V_Rf   ]
  2: 7           Y Loc [ V_Rf   ]
  3: 3           Z Loc [ cal    ]

16: Z=X*F (P37)
  1: 3           X Loc [ cal    ]
  2: 128.7       F
  3: 3           Z Loc [ cal    ]

17: Z=X+Y (P33)
  1: 4           X Loc [ mV_0   ]
  2: 6           Y Loc [ mV_end  ]
  3: 9           Z Loc [ work   ]

18: Z=X*F (P37)
  1: 9           X Loc [ work   ]
  2: .5          F
  3: 9           Z Loc [ work   ]

19: Z=X-Y (P35)
  1: 9           X Loc [ work   ]
  2: 5           Y Loc [ mV_180 ]
  3: 9           Z Loc [ work   ]

20: Z=ABS(X) (P43)
  1: 9           X Loc [ work   ]
  2: 9           Z Loc [ work   ]

21: Z=X/Y (P38)
  1: 3           X Loc [ cal    ]
  2: 9           Y Loc [ work   ]
  3: 3           Z Loc [ cal    ]

22: End (P95)

23: End (P95)

End Program

-Input Locations-
1 shf
2 shf_mV
3 cal
4 mV_0
5 mV_180
6 mV_end
7 V_Rf
8 shf_mV_a
9 work

```

<b>TABLE 5. Wiring for Example 3</b>		
Description	Color	CR5000
Sensor Signal #1	White	11H
Sensor Signal Reference #1	Green	11L
Shield #1	Clear	⚡
Sensor Signal #2	White	12H
Sensor Signal Reference #2	Green	12L
Shield #2	Clear	⚡
Sensor Signal #3	White	13H
Sensor Signal Reference #3	Green	13L
Shield #3	Clear	⚡
Sensor Signal #4	White	14H
Sensor Signal Reference #4	Green	14L
Shield #4	Clear	⚡
Heater Resistor Signal #1	Yellow	17H
Heater Resistor Signal Reference #1	Purple	17L
Shield #1	Clear	⚡
Power #1	Red	SW12
Power Reference #1	Black	G
Heater Resistor Signal #2	Yellow	18H
Heater Resistor Signal Reference #2	Purple	18L
Shield #2	Clear	⚡
Power #2	Red	SW12
Power Reference #2	Black	G
Heater Resistor Signal #3	Yellow	19H
Heater Resistor Signal Reference #3	Purple	19L
Shield #3	Clear	⚡
Power #3	Red	SW12
Power Reference #3	Black	G
Heater Resistor Signal #4	Yellow	20H
Heater Resistor Signal Reference #4	Purple	20L
Shield #4	Clear	⚡
Power #4	Red	SW12

**Example 3. Sample CR5000 Program Using a Differential Measurement Instruction**

```
'CR5000 Series Datalogger

Const OUTPUT_INTERVAL = 30      'Online mean output interval in minutes.
Const CAL_INTERVAL = 1440      'HFP01SC insitu calibration interval (minutes).
Const END_CAL = OUTPUT_INTERVAL-1 'End HFP01SC insitu calibration one minute before the next Output.
Const HFP01SC_CAL_1 = 15      'Unique multiplier for HFP01SC #1 (1000/sensitivity).
Const HFP01SC_CAL_2 = 15      'Unique multiplier for HFP01SC #2 (1000/sensitivity).
Const HFP01SC_CAL_3 = 15      'Unique multiplier for HFP01SC #3 (1000/sensitivity).
Const HFP01SC_CAL_4 = 15      'Unique multiplier for HFP01SC #4 (1000/sensitivity).

*** Variables ***

Public shf(4)
Alias shf(1) = hfp01sc_1
Alias shf(2) = hfp01sc_2
Alias shf(3) = hfp01sc_3
Alias shf(4) = hfp01sc_4
Units shf = W/m^2

Public shf_cal(4)
Units shf_cal = W/(m^2 mV)

'HFP01SC calibration variables.
Dim shf_mV(4)
Dim shf_mV_run(4)
Dim shf_mV_0(4)
Dim shf_mV_180(4)
Dim shf_mV_end(4)
Dim V_Rf(4)
Dim V_Rf_run(4)
Dim V_Rf_180(4)
Dim shf_cal_on
Dim sw12_state      'State of the switched 12Vdc port.
Dim j

DataTable (mean,TRUE,100)
  DataInterval (0,OUTPUT_INTERVAL,Min,10)

  Average (4,hfp01sc_1,IIEEE4,shf_cal_on)
  Sample (4,shf_cal(1),IIEEE4)
EndTable

Sub hfp01sc_cal
  'Begin HFP01SC calibration on a fixed interval.
  If ( IfTime (1,CAL_INTERVAL,Min) ) Then
    shf_cal_on = TRUE
    Move (shf_mV_0(1),4,shf_mV_run(1),4)
    sw12_state = TRUE
  EndIf
```

```

If ( IfTime (4,CAL_INTERVAL,Min) ) Then
  Move (shf_mV_180(1),4,shf_mV_run(1),4)
  Move (V_Rf_180(1),4,V_Rf_run(1),4)
  sw12_state = FALSE
EndIf

If ( IfTime (END_CAL,CAL_INTERVAL,Min) ) Then
  Move (shf_mV_end(1),4,shf_mV_run(1),4)
  'Compute new HFP01SC calibration factors.
  For j = 1 to 4
    shf_cal(j) = (V_Rf_180(j)*V_Rf_180(j)*128.7)/ABS (((shf_mV_0(j)+shf_mV_end(j))/2)-shf_mV_180)
  Next j
  shf_cal_on = FALSE
EndIf
EndSub

BeginProg
  'HFP01SC factory calibration in W/(m^2 mV) = 1000/sensitivity.
  shf_cal(1) = HFP01SC_CAL_1
  shf_cal(2) = HFP01SC_CAL_2
  shf_cal(3) = HFP01SC_CAL_3
  shf_cal(4) = HFP01SC_CAL_4

  Scan (100,mSec,3,0)
  'Measure the HFP01SC soil heat flux plates.
  VoltDiff (shf_mV(1),4,mV50C,11,TRUE,200,250,1,0)

  'Apply calibration to HFP01SC soil heat flux plates.
  For j = 1 to 4
    shf(j) = shf_mV(j)*shf_cal(j)
  Next j

  'Measure voltage across the heater (Rf_V).
  VoltSe (V_Rf(1),4,mV5000,33,TRUE,200,250,0.001,0)

  'Maintain a 100 sample running average.
  AvgRun (shf_mV_run(1),4,shf_mV(1),100)
  AvgRun (V_Rf_run(1),4,V_Rf(1),100)

  CallTable (mean)
  Call hfp01sc_cal
NextScan
EndProg

```

<b>TABLE 6A. Wiring for Example 4</b>			
Description	Color	CR10X	A21REL-12
Sensor Signal #1	White	1H	
Sensor Signal #2	White	1L	
Sensor Signal #3	White	2H	
Sensor Signal #4	White	2L	
Sensor Signal #5	White	3H	
Sensor Signal #6	White	3L	
All Signal References	Green	AG	
All Shields	Clear	G	
Heater Resistor Signal #1	Yellow	4H	
Heater Resistor Signal #2	Yellow	4L	
Heater Resistor Signal #3	Yellow	5H	
Heater Resistor Signal #4	Yellow	5L	
Heater Resistor Signal #5	Yellow	6H	
Heater Resistor Signal #6	Yellow	6L	
All Heater Resistor Signal References	Purple	AG	
All Shields	Clear	G	
Sensor Power #1	Red		REL 1 NO
Sensor Power #2	Red		REL 1 NO
Sensor Power #3	Red		REL 2 NO
Sensor Power #4	Red		REL 2 NO
Sensor Power #5	Red		REL 3 NO
Sensor Power #6	Red		REL 3 NO
All Power Reference	Black	G	

<b>TABLE 6B. Wiring for Example 4</b>		
Description	CR10X	A21REL-12
Power	12V	+ 12V
Power Reference	G	GROUND
Control	C8	CTRL 1
Control		jumper from CTRL 2 to CTRL 1
Control		jumper from CTRL 3 to CTRL 2
Power		jumper from REL 1 COM to +12V
Power		jumper from REL 2 COM to REL 1 COM
Power		jumper for REL 3 COM to REL 2 COM

**Example 4. Sample CR10X Program Using External Power and Relay**

```

;{CR10X}
;
*Table 1 Program
 01: 1          Execution Interval (seconds)

;Measure HFP01SC on smallest range.
;
1: Volt (SE) (P1)
 1: 6          Reps
 2: 22         7.5 mV 60 Hz Rejection Range
 3: 1          SE Channel
 4: 7          Loc [ shf_mV_1 ]
 5: 1          Mult
 6: 0          Offset

;Measure HFP01SC on larger range.
;
2: Volt (SE) (P1)
 1: 6          Reps
 2: 23         25 mV 60 Hz Rejection Range
 3: 1          SE Channel
 4: 44         Loc [ shf_mV_1a ]
 5: 1          Mult
 6: 0          Offset

;Load in the factory calibration.
;
3: If (X<=>F) (P89)
 1: 13         X Loc [ cal_1 ]
 2: 1          =
 3: 0          F
 4: 30         Then Do

;Factory calibration in  $W/(m^2 mV) = 1000/sensitivity$ .
;
4: Z=F (P30)
 1: 1          F ;<- Enter the unique calibration for plate 1 here.
 2: 0          Exponent of 10
 3: 13         Z Loc [ cal_1 ]

5: Z=F (P30)
 1: 1          F ;<- Enter the unique calibration for plate 2 here.
 2: 0          Exponent of 10
 3: 14         Z Loc [ cal_2 ]

6: Z=F (P30)
 1: 1          F ;<- Enter the unique calibration for plate 3 here.
 2: 0          Exponent of 10
 3: 15         Z Loc [ cal_3 ]

```

```

7: Z=F (P30)
  1: 1      F                               ;<- Enter the unique calibration for plate 4 here.
  2: 0      Exponent of 10
  3: 16     Z Loc [ cal_4 ]

8: Z=F (P30)
  1: 1      F                               ;<- Enter the unique calibration for plate 5 here.
  2: 0      Exponent of 10
  3: 17     Z Loc [ cal_5 ]

9: Z=F (P30)
  1: 1      F                               ;<- Enter the unique calibration for plate 6 here.
  2: 0      Exponent of 10
  3: 18     Z Loc [ cal_6 ]

10: End (P95)

11: Beginning of Loop (P87)
  1: 0      Delay
  2: 6      Loop Count

;Use data from the larger measurement range.
;
12: If (X<=>F) (P89)
  1: 7 --   X Loc [ shf_mV_1 ]
  2: 4      <
  3: -99990 F
  4: 30     Then Do

13: Z=X (P31)
  1: 44 --  X Loc [ shf_mV_1a ]
  2: 7 --   Z Loc [ shf_mV_1 ]

14: End (P95)

;Apply custom calibration to raw soil heat flux measurement.
;
15: Z=X*Y (P36)
  1: 7 --   X Loc [ shf_mV_1 ]
  2: 13 --  Y Loc [ cal_1 ]
  3: 1 --   Z Loc [ shf_1 ]

16: End (P95)

;Output data.
;
17: If time is (P92)
  1: 0      Minutes (Seconds --) into a
  2: 20     Interval (same units as above)
  3: 10     Set Output Flag High (Flag 0)

18: Real Time (P77)^25251
  1: 0110   Day,Hour/Minute (midnight = 0000)

```

```

19: Resolution (P78)
   1: 1          High Resolution

;Do not include that calibration data in the soil heat flux.
;
20: If Flag/Port (P91)
   1: 18         Do if Flag 8 is High
   2: 19         Set Intermed. Proc. Disable Flag High (Flag 9)

21: Average (P71)^21989
   1: 6          Reps
   2: 1          Loc [ shf_1  ]

22: Do (P86)
   1: 29         Set Intermed. Proc. Disable Flag Low (Flag 9)

23: Sample (P70)^21779
   1: 6          Reps
   2: 13         Loc [ cal_1  ]

;Add other processing here.

;Call calibration routine.
;
24: Do (P86)
   1: 8          Call Subroutine 8

*Table 2 Program
  02: 0          Execution Interval (seconds)

*Table 3 Subroutines

;Calibration routine.
;
1: Beginning of Subroutine (P85)
   1: 8          Subroutine 8

;Perform in-situ calibration.
;
2: If time is (P92)
   1: 1          Minutes (Seconds --) into a
   2: 180        Interval (same units as above)
   3: 30         Then Do

3: Beginning of Loop (P87)
   1: 0          Delay
   2: 6          Loop Count

4: Z=X (P31)
   1: 7 --       X Loc [ shf_mV_1  ]
   2: 19 --      Z Loc [ mV_0_1  ]

5: End (P95)

```

```

;Begin heating for calibration.
;
6: Do (P86)
  1: 48          Set Port 8 High

;Used to filter data during and after calibration.
;
7: Do (P86)
  1: 18          Set Flag 8 High

8: End (P95)

;End site calibration three minutes after calibration started.
;
9: If time is (P92)
  1: 4           Minutes (Seconds --) into a
  2: 180        Interval (same units as above)
  3: 30         Then Do

;Measure voltage across current shunt resistor during calibration.
;This measurement is used to compute power.
;
10: Volt (SE) (P1)
  1: 6           Reps
  2: 25          2500 mV 60 Hz Rejection Range
  3:             SE Channel
  4: 37          Loc [ V_Rf_1 ]
  5: .001        Mult
  6: 0           Offset

11: Beginning of Loop (P87)
  1: 0           Delay
  2: 6           Loop Count

12: Z=X (P31)
  1: 7 --        X Loc [ shf_mV_1 ]
  2: 25 --       Z Loc [ mV_180_1 ]

13: End (P95)

;Turn off the soil heat flux plate heaters.
;
14: Do (P86)
  1: 58          Set Port 8 Low

15: End (P95)

;Compute in-situ calibration.
;
16: If time is (P92)
  1: 39          Minutes (Seconds --) into a
  2: 180        Interval (same units as above)
  3: 30         Then Do

```

```

17: Do (P86)
  1: 28          Set Flag 8 Low

18: Beginning of Loop (P87)
  1: 0          Delay
  2: 6          Loop Count

19: Z=X (P31)
  1: 7 --      X Loc [ shf_mV_1 ]
  2: 31 --     Z Loc [ mV_end_1 ]

20: Z=X*Y (P36)
  1: 37 --     X Loc [ V_Rf_1 ]
  2: 37 --     Y Loc [ V_Rf_1 ]
  3: 13 --     Z Loc [ cal_1 ]

21: Z=X*F (P37)
  1: 13 --     X Loc [ cal_1 ]
  2: 128.7     F
  3: 13 --     Z Loc [ cal_1 ]

22: Z=X+Y (P33)
  1: 19 --     X Loc [ mV_0_1 ]
  2: 31 --     Y Loc [ mV_end_1 ]
  3: 43        Z Loc [ work ]

23: Z=X*F (P37)
  1: 43        X Loc [ work ]
  2: .5        F
  3: 43        Z Loc [ work ]

24: Z=X-Y (P35)
  1: 43        X Loc [ work ]
  2: 25 --     Y Loc [ mV_180_1 ]
  3: 43        Z Loc [ work ]

25: Z=ABS(X) (P43)
  1: 43        X Loc [ work ]
  2: 43        Z Loc [ work ]

26: Z=X/Y (P38)
  1: 13 --     X Loc [ cal_1 ]
  2: 43        Y Loc [ work ]
  3: 13 --     Z Loc [ cal_1 ]

27: End (P95)

28: End (P95)

29: End (P95)

End Program

```

```

-Input Locations-
1 shf_1  1 1 1
2 shf_2  0 0 0
3 shf_3  0 0 0
4 shf_4  0 0 0
5 shf_5  0 0 0
6 shf_6  0 0 0
7 shf_mV_1 1 5 2
8 shf_mV_2 1 0 1
9 shf_mV_3 1 0 1
10 shf_mV_4 1 0 1
11 shf_mV_5 1 0 1
12 shf_mV_6 1 0 1
13 cal_1  5 5 3
14 cal_2  9 0 1
15 cal_3  9 0 1
16 cal_4  9 0 1
17 cal_5  9 0 1
18 cal_6  9 0 1
19 mV_0_1  9 1 1
20 mV_0_2  1 0 0
21 mV_0_3  0 0 0
22 mV_0_4  0 0 0
23 mV_0_5  0 0 0
24 mV_0_6  0 0 0
25 mV_180_1 1 1 1
26 mV_180_2 0 0 0
27 mV_180_3 0 0 0
28 mV_180_4 0 0 0
29 mV_180_5 0 0 0
30 mV_180_6 0 0 0
31 mV_end_1 1 1 1
32 mV_end_2 0 0 0
33 mV_end_3 0 0 0
34 mV_end_4 0 0 0
35 mV_end_5 0 0 0
36 mV_end_6 0 0 0
37 V_Rf_1  5 2 1
38 V_Rf_2  9 0 1
39 V_Rf_3  9 0 1
40 V_Rf_4  9 0 1
41 V_Rf_5  9 0 1
42 V_Rf_6  17 0 1
43 work  1 4 4
44 shf_mV_1a 5 1 1
45 shf_mV_2a 9 0 1
46 shf_mV_3a 9 0 1
47 shf_mV_4a 9 0 1
48 shf_mV_5a 9 0 1
49 shf_mV_6a 17 0 1
    
```

## 6. Soil Heat Flux and Storage

The soil heat flux at the surface is calculated by adding the measured flux at a fixed depth,  $d$ , to the energy stored in the layer above the heat flux plates. The specific heat of the soil and the change in soil temperature,  $\Delta T_s$ , over the output interval,  $t$ , are required to calculate the stored energy.

The heat capacity of the soil is calculated by adding the specific heat of the dry soil to that of the soil water. The values used for specific heat of dry soil and water are on a mass basis. The heat capacity of the moist is given by:

$$C_s = \rho_b(C_d + \theta_m C_w) = \rho_b C_d + \theta_v \rho_w C_w \quad (1)$$

$$\theta_m = \frac{\rho_w}{\rho_b} \theta_v \quad (2)$$

where  $C_s$  is the heat capacity of moist soil,  $\rho_b$  is bulk density,  $\rho_w$  is the density of water,  $C_d$  is the heat capacity of a dry mineral soil,  $\theta_m$  is soil water content on a mass basis,  $\theta_v$  is soil water content on a volume basis, and  $C_w$  is the heat capacity of water.

This calculation requires site specific inputs for bulk density, mass basis soil water content or volume basis soil water content, and the specific heat of the dry soil. Bulk density and mass basis soil water content can be found by sampling (Klute, 1986). The volumetric soil water content is measured by the CS616 water content reflectometer. A value of  $840 \text{ J kg}^{-1} \text{ K}^{-1}$  for the heat capacity of dry soil is a reasonable value for most mineral soils (Hanks and Ashcroft, 1980).

The storage term is then given by Eq. (3) and the soil heat flux at the surface is given by Eq. (4).

$$S = \frac{\Delta T_s C_s d}{t} \quad (3)$$

$$G_{sfc} = G_{8cm} + S \quad (4)$$

## 7. Maintenance

The HFP01SC requires minimal maintenance. Check the sensor leads monthly for rodent damage.

## 8. In-Situ Calibration Theory

For detailed information on the theory of the in-situ calibration, see the Theory section of the Hukseflux manual or visit the application section of the Hukseflux web site at <http://www.hukseflux.com/heat%20flux/applic&spec.pdf>.

Equation 6 in the Hukseflux manual is used to compute a new calibration every three hours. The heater is on for a total of 180 seconds. Table 7 lists the

variables used in the Hukseflux manual and those in the example datalogger programs.

**TABLE 7. Hukseflux and Campbell Scientific Variable Names**

Description	Hukseflux	Campbell Scientific
Soil Heat Flux	$\phi$	shf
Output of Sensor in mV	$V_{sen}$	shf_mV
1/Sensitivity	$1/E_{sen2}$	cal
Output of Sensor during calibration at t=0 seconds	V (0)	mV_0
Output of Sensor during calibration at t=180 seconds	V (180)	mV_180
Output of Sensor after calibration and just before output	V (360)	mV_end
Voltage Across fixed 10 $\Omega$ resistor	$V_{cur}$	V_Rf

## 9. References

- Hanks, R. J., and G. L. Ashcroft, 1980: *Applied Soil Physics: Soil Water and Temperature Application*. Springer-Verlag, 159 pp.
- Klute, A., 1986: *Method of Soil Analysis*. No. 9, Part 1, Sections 13 and 21, American Society of Agronomy, Inc., Soil Science Society of America, Inc.





## **Campbell Scientific Companies**

---

### **Campbell Scientific, Inc. (CSI)**

815 West 1800 North  
Logan, Utah 84321  
UNITED STATES  
www.campbellsci.com  
info@campbellsci.com

### **Campbell Scientific Africa Pty. Ltd. (CSAf)**

PO Box 2450  
Somerset West 7129  
SOUTH AFRICA  
www.csafrica.co.za  
cleroux@csafrica.co.za

### **Campbell Scientific Australia Pty. Ltd. (CSA)**

PO Box 444  
Thuringowa Central  
QLD 4812 AUSTRALIA  
www.campbellsci.com.au  
info@campbellsci.com.au

### **Campbell Scientific do Brazil Ltda. (CSB)**

Rua Luisa Crapsi Orsi, 15 Butantã  
CEP: 005543-000 São Paulo SP BRAZIL  
www.campbellsci.com.br  
suporte@campbellsci.com.br

### **Campbell Scientific Canada Corp. (CSC)**

11564 - 149th Street NW  
Edmonton, Alberta T5M 1W7  
CANADA  
www.campbellsci.ca  
dataloggers@campbellsci.ca

### **Campbell Scientific Ltd. (CSL)**

Campbell Park  
80 Hathern Road  
Shepshed, Loughborough LE12 9GX  
UNITED KINGDOM  
www.campbellsci.co.uk  
sales@campbellsci.co.uk

### **Campbell Scientific Ltd. (France)**

Miniparc du Verger - Bat. H  
1, rue de Terre Neuve - Les Ulis  
91967 COURTABOEUF CEDEX  
FRANCE  
www.campbellsci.fr  
info@campbellsci.fr

### **Campbell Scientific Spain, S. L.**

Psg. Font 14, local 8  
08013 Barcelona  
SPAIN  
www.campbellsci.es  
info@campbellsci.es